Thanks for inviting me to Geneva!

ωεβ:=http://drorbn.net/ge24

Knot Invariants from Finite Dimensional Integration

Abstract. For the purpose of today, an "I-Type Knot Invariant" is a knot invariant computed from a knot diagram by integrating the exponential of a *perturbed Gaussian* Lagrangian which is a sum over the features of that diagram (crossings, edges, faces) of locally defined quantities, over a product of finite dimensional spaces associated to those same features.

joint with R. van der Veen

Q. Are there any such things? **A.** Yes.

Q. Are they any good? **A.** They are the strongest we know per CPU cycle, and are excellent in other ways too.

Q. Didn't Witten do that back in 1988 with path integrals?

A. No. His constructions are infinite dimensional and far from rigorous.

Q. But integrals belong in analysis!

A. Ours only use squeaky-clean algebra.

Knots.



Something simple: numbers, polynomials, matrices, etc.

The Good. 1. At the centre of low dimensional topology.
2. "Invariants" connect to pretty much all of algebra.

The Agony. 1&2 don't talk to each other.

- Not enough topological applications for all these invariants.
- The fancy algebra doesn't arise naturally within topology.

$\Longrightarrow$ We're still missing something about the relationship between knots and algebra.

The $sl_2^{/\epsilon^2}$ Example. With T an indeterminate and with $\epsilon^2=0$:



$$\longrightarrow Z=\oint_{\mathbb{R}^{14}_{p_ix_i}}\mathcal{L}(X^+_{15})\mathcal{L}(X^+_{62})\mathcal{L}(X^+_{37})\mathcal{L}(C_4^{-1})$$

measure on $\mathbb{R}$ is $(2\pi)^{-1/2}\cdot$*standard*

where $\mathcal{L}(X^s_{ij})=T^{s/2}\mathbb{e}^{L(X^s_{ij})}$ and $\mathcal{L}(C^\varphi_i)=T^{\varphi/2}\mathbb{e}^{L(C^\varphi_i)}$, and

$$L(X^s_{ij})=x_i(p_{i+1}-p_i)+x_j(p_{j+1}-p_j)+(T^s-1)x_i(p_{i+1}-p_{j+1})+\frac{\epsilon s}{2}\left(x_i(p_i-p_j)\begin{pmatrix}(T^s-1)x_ip_j\\+2(1-x_jp_j)\end{pmatrix}-1\right)$$

$$L(C^\varphi_i)=x_i(p_{i+1}-p_i)+\epsilon\varphi(1/2-x_ip_i)$$

So $Z=T\oint\mathbb{e}^{L(\text{knot})}dp_1\dots dp_7dx_1\dots dx_7$, where $L(\text{knot})=\sum_{i=1}^7x_i(p_{i+1}-p_i)+(T-1)(x_1(p_2-p_6)+x_6(p_7-p_3)+x_3(p_4-p_8))$

$$+\frac{\epsilon}{2}\begin{pmatrix}x_1(p_1-p_5)\left((T-1)x_1p_5+2(1-x_5p_5)\right)-1\\+x_6(p_6-p_2)\left((T-1)x_6p_2+2(1-x_2p_2)\right)-1\\+x_3(p_3-p_7)\left((T-1)x_3p_7+2(1-x_7p_7)\right)-1\\+2x_4p_4-1\end{pmatrix},$$

and so $Z=(T-1+T^{-1})^{-1}\exp\left(\epsilon\cdot\frac{(T-2+T^{-1})(T+T^{-1})}{(T-1+T^{-1})^2}\right)=\Delta^{-1}\exp\left(\epsilon\cdot\frac{(T-2+T^{-1})\rho_1}{\Delta^2}\right)$. Here Δ is Alexander's polynomial and ρ_1 is Rozansky-Overbay's polynomial [R1, R2, R3, Ov, BV1, BV2].



Rozansky

Overbay

Theorem. Z is a knot invariant. **Proof.** Use Fubini (details later).



(Alternative) Gaussian Integration.



Gauss

Goal. Compute (if convergent) $\displaystyle\int_{\mathbb{R}^n}dx\exp\left(-\frac{1}{2}a^{ij}x_ix_j+V(x)\right)$.

Solution. Set $\displaystyle\mathcal{Z}_\lambda(x):=\lambda^{n/2}\int_{\mathbb{R}^n}dy\exp\left(-\frac{1}{2\lambda}a^{ij}y_iy_j+V(x+y)\right)$.

Then $\mathcal{Z}_1(0)$ is what we want, $\mathcal{Z}_0(x)=(\det A)^{-1/2}\exp V(x)$, and with g_{ij} the inverse matrix of a^{ij} and noting that under the dy integral $\partial_y=0$,

$$\frac{1}{2}g_{ij}\partial_{x_i}\partial_{x_j}\mathcal{Z}_\lambda(x)$$
$$=\frac{1}{2}\int_{\mathbb{R}^n}dy\,g_{ij}(\partial_{x_i}-\partial_{y_i})(\partial_{x_j}-\partial_{y_j})\exp\left(-\frac{1}{2\lambda}a^{ij}y_iy_j+V(x+y)\right)$$
$$=\frac{1}{2\lambda^2}\int_{\mathbb{R}^n}dy\left(g_{ij}a^{ii'}a^{jj'}y_{i'}y_{j'}+\lambda g_{ij}a^{ji}\right)\exp\left(-\frac{1}{2\lambda}a^{ij}y_iy_j+V(x+y)\right)$$
$$=\frac{1}{2\lambda^2}\int_{\mathbb{R}^n}dy\left(a^{ij}y_iy_j+\lambda n\right)\exp\left(-\frac{1}{2\lambda}a^{ij}y_iy_j+V(x+y)\right)$$
$$=\partial_\lambda\mathcal{Z}_\lambda(x).$$

Hence
$$(*)\quad\partial_\lambda\mathcal{Z}_\lambda(x)=\frac{1}{2}g_{ij}\partial_{x_i}\partial_{x_j}\mathcal{Z}_\lambda(x),$$
and therefore
$$\mathcal{Z}_\lambda(x)=(\det A)^{-1/2}\exp\left(\frac{\lambda}{2}g_{ij}\partial_{x_i}\partial_{x_j}\right)\exp V(x).$$

We've just witnessed the birth of "Feynman Diagrams".



Feynman

Even better. With $Z_\lambda:=\log(\sqrt{\det A}\mathcal{Z}_\lambda)$, by a simple substitution into (*), we get the "Synthesis Equation":

$$Z_0=V,\quad\partial_\lambda Z_\lambda=\frac{1}{2}\sum_{i,j=1}^n g_{ij}\left(\partial_{x_i,x_j}Z_\lambda+(\partial_{x_i}Z_\lambda)(\partial_{x_j}Z_\lambda)\right)=:F(Z_\lambda),$$

an ODE (in λ) whose solution is pure algebra.

Picard Iteration (used to prove the existence and uniqueness of solutions of ODEs). To solve $\partial_\lambda f_\lambda=F(f_\lambda)$ with a given f_0, start with f_0, iterate $f\mapsto f_0+\int_0^\lambda F(f_\lambda)d\lambda$, and seek a fixed point. In our cases, it is always reached after finitely many iterations!



Picard

Definition. $\oint$: The result of this process, ignoring the convergence of the actual integral.

Strong. The pair (Δ,ρ_1) attains 53,684 distinct values on the 59,937 prime knots with up to 14 crossings (a deficit of 6,253), whereas the pair (H = HOMFLYPT polynomial, Kh = Khovanov Homology) attains only 49,149 distinct values on the same knots (a deficit of 10,788). The pair (Δ,θ), discussed later, has a deficit of only 1,118.

Yet better than (H,Kh) and other Reshetikhin-Turaev-Witten invariants and knot homologies, Δ, ρ_1, and θ can be computed in **polynomial time** (and hence, even for very large knots).

So ugly as the formulas may be (and θ's formulas are uglier), these invariants are **the best we have!**

Acknowledgement. This work was supported by NSERC grant RGPIN-2018-04350 and by the Chu Family Foundation (NYC).

Implementation (see IType.nb of ωεβ/ap).

```
☺ Once[<< KnotTheory`; << Rot.m];
💻 C:\drorbn\AcademicPensieve\Projects\KnotTheory\KnotTheory
💻 Loading KnotTheory` version
     of February 2, 2020, 10:53:45.2097.
   Read more at http://katlas.org/wiki/KnotTheory.
💻 Loading Rot.m from
     http://drorbn.net/AP/Talks/Geneva-2408
     to compute rotation numbers.
```

```
☺ CF[ω_. ℰ_𝔼] := CF[ω] × CF /@ ℰ;
  CF[ℰ_List] := CF /@ ℰ;
  CF[ℰ_] := Module[{vs, ps, c},
     vs = Cases[ℰ, (x | p | ξ | π | g)__, ∞] ⋃ {ϵ};
     Total[CoefficientRules[Expand[ℰ], vs] /.
       (ps_ → c_) ⧴ Factor[c] (Times @@ vs^ps)] ];
```

Integration using Picard iteration. The core is in yellow and hacks are in pink.

```
☺ 𝔼 /: 𝔼[A_] × 𝔼[B_] := 𝔼[A + B];
☺ $π = Identity; (* The Wisdom Projection *)
☺ Unprotect[Integrate];
  ∫ ω_. 𝔼[L_] ⅆ(vs_List) :=
   Module[{n, L0, Q, Δ, G, Z0, Z, λ, DZ, DDZ, FZ,
      a, b},
     n = Length@vs; L0 = L /. ϵ → 0;
     Q = Table[(-∂_{vs⟦a⟧,vs⟦b⟧} L0) /. Thread[vs → 0] /.
        (p | x)__ → 0, {a, n}, {b, n}];
     If[(Δ = Det[Q]) == 0, Return@"Degenerate Q!"];
     Z = Z0 = CF@$π[L + vs.Q.vs / 2]; G = Inverse[Q];
     FixedPoint[(DZ = Table[∂_v Z, {v, vs}];
        DDZ = Table[∂_u DZ, {u, vs}];
        FZ = Sum[G⟦a, b⟧ (DDZ⟦a, b⟧ + DZ⟦a⟧ × DZ⟦b⟧),
           {a, n}, {b, n}] / 2;
        Z = CF[Z0 + ∫_0^λ $π[FZ] ⅆλ]) &, Z];
     PowerExpand@Factor[ω Δ^(-1/2)] ×
      𝔼[CF[Z /. λ → 1 /. Thread[vs → 0]]]];
  Protect[Integrate];
```

☺ $\int \mathbb{E}[-\mu x^2/2 + i\,\xi\,x]\, d\{x\}$

💻 $\dfrac{\mathbb{E}\left[-\frac{\xi^2}{2\mu}\right]}{\sqrt{\mu}}$

☺ FofG $= \int \mathbb{E}[-\mu\,(x-a)^2/2 + i\,\xi\,x]\, d\{x\}$

💻 $\dfrac{\mathbb{E}\left[\frac{i\,(2\,a\,\mu + i\,\xi)\,\xi}{2\,\mu}\right]}{\sqrt{\mu}}$

Joseph Fourier

☺ $\int$ FofG $\mathbb{E}[-i\,\xi\,x]\, d\{\xi\}$

💻 $\mathbb{E}\left[-\frac{1}{2}\,(a-x)^2\,\mu\right]$

So we've tested and nearly proven the Fourier inversion formula!

☺ $L = -\frac{1}{2}\{x_1, x_2\}.\begin{pmatrix} a & b \\ b & c\end{pmatrix}.\{x_1, x_2\} + \{\xi_1, \xi_2\}.\{x_1, x_2\};$

Z12 $= \int \mathbb{E}[L]\, d\{x_1, x_2\}$

💻 $\dfrac{\mathbb{E}\left[\frac{c\,\xi_1^2}{2\,(-b^2+a\,c)} + \frac{b\,\xi_1\,\xi_2}{b^2-a\,c} + \frac{a\,\xi_2^2}{2\,(-b^2+a\,c)}\right]}{\sqrt{-b^2+a\,c}}$

☺ $\left\{\text{Z1} = \int \mathbb{E}[L]\, d\{x_1\},\ \text{Z12} == \int \text{Z1}\, d\{x_2\}\right\}$

💻 $\left\{\dfrac{\mathbb{E}\left[-\frac{(-b^2+a\,c)\,x_2^2}{2\,a} - \frac{b\,x_2\,\xi_1}{a} + \frac{\xi_1^2}{2\,a} + x_2\,\xi_2\right]}{\sqrt{a}},\ \text{True}\right\}$

Guido Fubini

☺ $\$\pi$ = Normal[# + O[ϵ]^13] &; $\int \mathbb{E}[-\phi^2/2 + \epsilon\,\phi^3/6]\, d\{\phi\}$

💻 $\mathbb{E}\left[\frac{5\,\epsilon^2}{24} + \frac{5\,\epsilon^4}{16} + \frac{1105\,\epsilon^6}{1152} + \frac{565\,\epsilon^8}{128} + \frac{82825\,\epsilon^{10}}{3072} + \frac{19675\,\epsilon^{12}}{96}\right]$

From https://oeis.org/A226260:

THE ON-LINE ENCYCLOPEDIA OF INTEGER SEQUENCES ®

founded in 1964 by N. J. A. Sloane

Search Hints

(Greetings from The On-Line Encyclopedia of Integer Sequences!)

A226260 Numerators of mass formula for connected vacuum graphs on 2n nodes for a phi^3 field theory.
1, 5, 5, 1105, 565, 82825, 19675, 1282031525, 80727925, 1683480621875, 13209845125, 2239646759308375, 19739117098375, 6320791709083309375, 32468078556378125, 38362676768845045751875, 281365778405032973125, 2824650747089425586152484375, 776632157034116712734375 (list; graph; refs; listen; history; text; internal format)

The Right-Handed Trefoil.

```
☺ K = Mirror@Knot[3, 1]; Features[K]
💻 Features[7, C4[-1] X1,5[1] X3,7[1] X6,2[1]]
☺ ℒ[X_{i_,j_}[s_]] := T^(s/2) 𝔼[
     x_i (p_{i+1} - p_i) + x_j (p_{j+1} - p_j) +
      (T^s - 1) x_i (p_{i+1} - p_{j+1}) +
      (ϵ s / 2) ×
       (x_i (p_i - p_j) ((T^s - 1) x_i p_j + 2 (1 - x_j p_j)) - 1)]
  ℒ[C_{i_}[φ_]] := T^(φ/2) 𝔼[x_i (p_{i+1} - p_i) + ϵ φ (1/2 - x_i p_i)]
  ℒ[K_] := CF[ℒ /@ Features[K]⟦2⟧]
  vs[K_] :=
   Join @@ Table[{p_i, x_i}, {i, Features[K]⟦1⟧}]
```

```
{vs[K], ℒ[K]}
```

$$\{p_1, x_1, p_2, x_2, p_3, x_3, p_4, x_4, p_5, x_5, p_6, x_6, p_7, x_7\},$$
$$T\,\mathbb{E}\Big[-2\epsilon - p_1 x_1 + \epsilon p_1 x_1 + T p_2 x_1 - \epsilon p_5 x_1 + (1-T) p_6 x_1 + \frac{1}{2}(-1+T)\epsilon p_1 p_5 x_1^2 + \frac{1}{2}(1-T)\epsilon p_5^2 x_1^2 - p_2 x_2 + p_3 x_2 - p_3 x_3 + \epsilon p_3 x_3 + T p_4 x_3 - \epsilon p_7 x_3 + (1-T) p_8 x_3 + \frac{1}{2}(-1+T)\epsilon p_3 p_7 x_3^2 + \frac{1}{2}(1-T)\epsilon p_7^2 x_3^2 - p_4 x_4 + \epsilon p_4 x_4 + p_5 x_4 - p_5 x_5 + p_6 x_5 - \epsilon p_1 p_5 x_1 x_5 + \epsilon p_5^2 x_1 x_5 - \epsilon p_2 x_6 + (1-T) p_3 x_6 - p_6 x_6 + \epsilon p_6 x_6 + T p_7 x_6 + \epsilon p_2^2 x_2 x_6 - \epsilon p_2 p_6 x_2 x_6 + \frac{1}{2}(1-T)\epsilon p_2^2 x_6^2 + \frac{1}{2}(-1+T)\epsilon p_2 p_6 x_6^2 - p_7 x_7 + p_8 x_7 - \epsilon p_3 p_7 x_3 x_7 + \epsilon p_7^2 x_3 x_7\Big]$$

```
$π = Normal[# + O[ϵ]^2] &; ∫ℒ[K] ⅆ vs[K]
```

$$-\frac{i\,T\,\mathbb{E}\left[-\frac{(-1+T)^2(1+T^2)\epsilon}{(1-T+T^2)^2}\right]}{1-T+T^2}$$

A faster program to compute ρ_1, and more stories about it, are at [BV2].

Invariance Under Reidemeister 3.




Reidemeister

```
lhs = ∫(ℒ /@ (X_{i,j}[1] X_{i+1,k}[1] X_{j+1,k+1}[1]))
    ⅆ{p_{i+1}, p_{j+1}, p_{k+1}, x_{i+1}, x_{j+1}, x_{k+1}};
rhs = ∫(ℒ /@ (X_{j,k}[1] X_{i,k+1}[1] X_{i+1,j+1}[1]))
    ⅆ{x_{i+1}, p_{i+1}, p_{j+1}, p_{k+1}, x_{j+1}, x_{k+1}};
lhs === rhs
```

```
False
```

Invariance Under Reidemeister 3, Take 2.

```
lhs = ∫(ℒ /@ (X_{i,j}[1] X_{i+1,k}[1] X_{j+1,k+1}[1]))
    ⅆ{x_i, x_j, x_k, p_{i+1}, p_{j+1}, p_{k+1}, x_{i+1}, x_{j+1}, x_{k+1}};
rhs = ∫(ℒ /@ (X_{j,k}[1] X_{i,k+1}[1] X_{i+1,j+1}[1]))
    ⅆ{x_i, x_j, x_k, x_{i+1}, p_{i+1}, p_{j+1}, p_{k+1}, x_{j+1}, x_{k+1}};
lhs === rhs
```

```
True
```

```
lhs
```

```
Degenerate Q!
```

Invariance Under Reidemeister 3, Take 3.

```
lhs = ∫(𝔼[ⅈ π_i p_i + ⅈ π_j p_j + ⅈ π_k p_k] × ℒ /@ (X_{i,j}[1] X_{i+1,k}[1] X_{j+1,k+1}[1]))
    ⅆ{p_i, p_j, p_k, x_i, x_j, x_k, p_{i+1}, p_{j+1}, p_{k+1}, x_{i+1}, x_{j+1}, x_{k+1}};
rhs = ∫(𝔼[ⅈ π_i p_i + ⅈ π_j p_j + ⅈ π_k p_k] × ℒ /@ (X_{j,k}[1] X_{i,k+1}[1] X_{i+1,j+1}[1]))
    ⅆ{p_i, p_j, p_k, x_i, x_j, x_k, p_{i+1}, p_{j+1}, p_{k+1}, x_{i+1}, x_{j+1}, x_{k+1}};
lhs == rhs
```

```
True
```

```
lhs
```

$$T^{3/2}\,\mathbb{E}\Big[-\frac{3\epsilon}{2} + i T^2 p_{2+i}\pi_i - i(-1+T)T p_{2+j}\pi_i + i T^2\epsilon p_{2+j}\pi_i - i(-1+T)p_{2+k}\pi_i + i T\epsilon p_{2+k}\pi_i - \frac{1}{2}(-1+T)T^3\epsilon p_{2+i}p_{2+j}\pi_i^2 + \frac{1}{2}(-1+T)T^3\epsilon p_{2+j}^2\pi_i^2 - \frac{1}{2}(-1+T)T^2\epsilon p_{2+i}p_{2+k}\pi_i^2 + \frac{1}{2}(-1+T)^2 T\epsilon p_{2+j}p_{2+k}\pi_i^2 + \frac{1}{2}(-1+T)T\epsilon p_{2+k}^2\pi_i^2 + i T p_{2+j}\pi_j - i T\epsilon p_{2+j}\pi_j - i(-1+T)p_{2+k}\pi_j + i(-1+2T)\epsilon p_{2+k}\pi_j + T^3\epsilon p_{2+i}p_{2+j}\pi_i\pi_j - T^3\epsilon p_{2+j}^2\pi_i\pi_j - (-1+T)T^2\epsilon p_{2+i}p_{2+k}\pi_i\pi_j + (-1+T)^2 T\epsilon p_{2+j}p_{2+k}\pi_i\pi_j + (-1+T)T\epsilon p_{2+k}^2\pi_i\pi_j - \frac{1}{2}(-1+T)T\epsilon p_{2+j}p_{2+k}\pi_j^2 + \frac{1}{2}(-1+T)T\epsilon p_{2+k}^2\pi_j^2 + i p_{2+k}\pi_k - 2i\epsilon p_{2+k}\pi_k + T^2\epsilon p_{2+i}p_{2+k}\pi_i\pi_k - (-1+T)T\epsilon p_{2+j}p_{2+k}\pi_i\pi_k - T\epsilon p_{2+k}^2\pi_i\pi_k + T\epsilon p_{2+j}p_{2+k}\pi_j\pi_k - T\epsilon p_{2+k}^2\pi_j\pi_k\Big]$$

Invariance under the other Reidemeister moves is proven in a similar way. See IType.nb at ωεβ/ap.

There's more! To get sl_2 invariants mod ϵ^3, add the following to $L(X_{ij}^+)$, $L(X_{ij}^-)$, and $L(C_i^\varphi)$, respectively (and see More.nb at ωεβ/ap for the verifications):

```
ϵ^2 r_2[1, i, j]
```

$$\frac{1}{12}\epsilon^2\big(-6 p_i x_i + 6 p_j x_i - 3(-1+3T)p_i p_j x_i^2 + 3(-1+3T)p_j^2 x_i^2 + 4(-1+T)p_i^2 p_j x_i^3 - 2(-1+T)(5+T)p_i p_j^2 x_i^3 + 2(-1+T)(3+T)p_j^3 x_i^3 + 18 p_i p_j x_i x_j - 18 p_j^2 x_i x_j - 6 p_i^2 p_j x_i^2 x_j + 6(2+T)p_i p_j^2 x_i^2 x_j - 6(1+T)p_j^3 x_i^2 x_j - 6 p_i p_j^2 x_i x_j^2 + 6 p_j^3 x_i x_j^2\big)$$

```
ϵ^2 r_2[-1, i, j]
```

$$\frac{1}{12T^2}\epsilon^2\big(-6T^2 p_i x_i + 6T^2 p_j x_i + 3(-3+T)T p_i p_j x_i^2 - 3(-3+T)T p_j^2 x_i^2 - 4(-1+T)T p_i^2 p_j x_i^3 + 2(-1+T)(1+5T)p_i p_j^2 x_i^3 - 2(-1+T)(1+3T)p_j^3 x_i^3 + 18T^2 p_i p_j x_i x_j - 18T^2 p_j^2 x_i x_j - 6T^2 p_i^2 p_j x_i^2 x_j + 6T(1+2T)p_i p_j^2 x_i^2 x_j - 6T(1+T)p_j^3 x_i^2 x_j - 6T^2 p_i p_j^2 x_i x_j^2 + 6T^2 p_j^3 x_i x_j^2\big)$$

```
ϵ^2 γ_2[φ, i]
```

$$-\frac{1}{2}\epsilon^2\varphi^2 p_i x_i$$

Even more! • The sl_2 formulas mod ϵ^4 are in the last page of the handout of [BN3].

- Using [GPV] we can show that every finite type invariant is I-Type.
- Probably, ⟨Reshetikhin-Turaev⟩ ⊂ ⟨I-Type⟩ efficiently.
- Possibly, ⟨Rozansky Polynomials⟩ ⊂ ⟨I-Type⟩ efficiently.
- Knot signatures are I-Type, at least mod 8.
- We already have some work on sl_3, and it leads to the strongest genuinely-computable knot invariant presently known.

The $sl_3^{/\epsilon^2}$ Example (continues Schaveling [Sch]). Here we have two formal variables T_1 and T_2, we set $T_3 := T_1T_2$, we integrate over 6 variables for each edge: p_{1i}, p_{2i}, p_{3i}, x_{1i}, x_{2i}, and x_{3i}.

Schaveling

```
T3 = T1 T2;  i_+ := i + 1;
$π =
  (CF@Normal[# + O[ϵ]^2] /.
     {π_is__ :> B^-1 π_is, x_is__ :> B^-1 x_is,
      p_is__ :> B p_is} /. ϵ B^b_ /; b < 0 -> 0 /. B -> 1) &;
```

```
vs_i_ := Sequence[p_{1,i}, p_{2,i}, p_{3,i}, x_{1,i}, x_{2,i}, x_{3,i}];
𝓕[is__] := 𝔼[Sum[π_{ν,i} p_{ν,i}, {i, {is}}, {ν, 3}]];
ℒ[K_] := CF[ℒ /@ Features[K][[2]]];
vs[K_] :=
 Union @@ Table[{vs_i}, {i, Features[K][[1]]}]
```

The Lagrangian.

```
ℒ[X_{i_,j_}[s_]] := T3^s 𝔼[CF@Plus[
   Σ_{ν=1}^3 (x_{νi} (p_{νi+} - p_{νi}) + x_{νj} (p_{νj+} - p_{νj}) + (T_ν^s - 1) x_{νi} (p_{νi+} - p_{νj+})),
   (T1^s - 1) p_{3j} x_{1i} (T2^s x_{2i} - x_{2j}),
   ϵ s (T3^s - 1) p_{1j} (p_{2i} - p_{2j}) x_{3i} / (T2^s - 1),
   ϵ s (1/2 + T2^s p_{1i} p_{2j} x_{1i} x_{2i} - p_{1i} p_{2j} x_{1i} x_{2j} - p_{3i} x_{3i} -
      (T2^s - 1) p_{2j} p_{3i} x_{2i} x_{3i} + (T3^s - 1) p_{2j} p_{3j} x_{2i} x_{3i} +
      2 p_{2j} p_{3i} x_{2j} x_{3i} + p_{1i} p_{3j} x_{1i} x_{3j} - p_{2i} p_{3j} x_{2i} x_{3j} -
      T2^s p_{2j} p_{3j} x_{2i} x_{3j} +
      ((T1^s - 1) p_{1j} x_{1i} (T2^{2s} p_{2j} x_{2i} - T2^s p_{2j} x_{2j} -
           (T2^s + 1) (T3^s - 1) p_{3j} x_{3i} + T2^s p_{3j} x_{3j}) +
         (T3^s - 1) p_{3j} x_{3i} (1 - T2^s p_{1i} x_{1i} + p_{2i} x_{2j} + (T2^s - 2) p_{2j} x_{2j})) /
      (T2^s - 1))]]
```

```
ℒ[C_{i_}[φ_]] := T3^φ 𝔼[Σ_{ν=1}^3 x_{νi} (p_{νi+} - p_{νi}) + ϵ φ (p_{3i} x_{3i} - 1/2)]
```

Reidemeister 3.

```
Short[
  lhs = ∫ 𝓕[i, j, k] × ℒ /@ (X_{i,j}[1] X_{i+,k}[1] X_{j+,k+}[1])
     d{vs_i, vs_j, vs_k, vs_{i+}, vs_{j+}, vs_{k+}}]
```

```
T1^3 T2^3
 𝔼[3ϵ/2 + T1^2 p_{1,2+i} π_{1,i} - (-1 + T1) T1 p_{1,2+j} π_{1,i} + <<150>>]
```

```
rhs = ∫ 𝓕[i, j, k] × ℒ /@ (X_{j,k}[1] X_{i,k+}[1] X_{i+,j+}[1])
    d{vs_i, vs_j, vs_k, vs_{i+}, vs_{j+}, vs_{k+}};
lhs == rhs
```

```
True
```

The Trefoil.

```
K = Knot[3, 1];  ∫ ℒ[K] d vs[K]
```

```
-((i T1^2 T2^2
   𝔼[-((ϵ (1 - T1 + T1^2 - T2 - T1^3 T2 + T2^2 + T1^4 T2^2 - T1 T2^3 -
          T1^4 T2^3 + T1^2 T2^4 - T1^3 T2^4 + T1^4 T2^4)) / ((1 - T1 + T1^2)
        (1 - T2 + T2^2) (1 - T1 T2 + T1^2 T2^2)))]) /
  ((1 - T1 + T1^2) (1 - T2 + T2^2) (1 - T1 T2 + T1^2 T2^2)))
```

A faster program, in which the Feynman diagrams are "pre-computed" (see theta.nb at ωεβ/ap):

```
R1[s_, i_, j_] = CF[
   s (1/2 - g_{3ii} + T2^s g_{1ii} g_{2ji} - g_{1ii} g_{2jj} - (T2^s - 1) g_{2ji} g_{3ii} +
      2 g_{2jj} g_{3ii} - (1 - T3^s) g_{2ji} g_{3ji} - g_{2ii} g_{3jj} - T2^s g_{2ji} g_{3jj} +
      g_{1ii} g_{3jj} +
      ((T1^s - 1) g_{1ji} (T2^{2s} g_{2ji} - T2^s g_{2jj} + T2^s g_{3jj}) +
         (T3^s - 1) g_{3ji} (1 - T2^s g_{1ii} - (T1^s - 1) (T2^s + 1) g_{1ji} +
            (T2^s - 2) g_{2jj} + g_{2ij})) / (T2^s - 1))];
```

```
θ[{s0_, i0_, j0_}, {s1_, i1_, j1_}] :=
  CF[s1 (T1^s0 - 1) (T2^s1 - 1)^-1 (T3^s1 - 1) g_{1,j1,i0} g_{3,j0,i1}
    ((T2^s0 g_{2,i1,i0} - g_{2,i1,j0}) - (T2^s0 g_{2,j1,i0} - g_{2,j1,j0}))]
```

```
Γ1[φ_, k_] = -φ/2 + φ g_{3kk};
```

We call the invariant computed θ:

```
θ[K_] := Module[{Cs, φ, n, A, s, i, j, k, Δ, G, ν, α, β, gEval, c, z},
   {Cs, φ} = Rot[K]; n = Length[Cs];
   A = IdentityMatrix[2 n + 1];
   Cases[Cs, {s_, i_, j_} :>
     (A[[{i, j}, {i + 1, j + 1}]] += ( -T^s  T^s - 1
                                        0     -1   ))];
   Δ = T^((-Total[φ] - Total[Cs[[All,1]]])/2) Det[A];
   G = Inverse[A];
   gEval[ℰ_] := Factor[ℰ /. g_{ν_,α_,β_} :> (G[[α, β]] /. T -> T_ν)];
   z = gEval[Σ_{k1=1}^n Σ_{k2=1}^n θ[Cs[[k1]], Cs[[k2]]]];
   z += gEval[Σ_{k=1}^n R1 @@ Cs[[k]]];
   z += gEval[Σ_{k=1}^{2n} Γ1[φ[[k]], k]];
   {Δ, (Δ /. T -> T1) (Δ /. T -> T2) (Δ /. T -> T3) z} // Factor];
```

Some Knots.

```
Expand[θ[Knot[3, 1]]]
```

$$\left\{-1+\frac{1}{T}+T,\ -\frac{1}{T_1^2}-T_1^2-\frac{1}{T_2^2}-\frac{1}{T_1^2T_2^2}+\frac{1}{T_1T_2^2}+\frac{1}{T_1^2T_2}+\frac{T_1}{T_2}+\frac{T_2}{T_1}+T_1^2T_2-T_2^2+T_1T_2^2-T_1^2T_2^2\right\}$$

```
PolyPlot[0] = Graphics[{}];
PolyPlot[p_] := Module[{crs, m1, m2, maxc, minc, s, hex},
   crs = CoefficientRules[T1^(m1=-Exponent[p,T1,Min]) T2^(m2=-Exponent[p,T2,Min]) p,
      {T1, T2}];
   maxc = N@Log@Max@Abs[Last /@ crs];
   minc = N@Log@Min@Select[Abs[Last /@ crs], # > 0 &];
   If[minc == maxc, s[_] = 0,
     s[c_] := s[c] = (maxc - Log@c) / (maxc - minc)];
   hex = Table[{Cos[α], Sin[α]} / Cos[2 π / 12] / 2,
      {α, 2 π / 12, 2 π, 2 π / 6}];
   Graphics[crs /. ({x1_, x2_} -> c_) :> {
      If[c == 0, White, Lighter[If[c > 0, Red, Blue],
        0.88 s[Abs@c]]],
      Polygon[(( 1  -1/2
                 0  √3/2 ).{x1 + m1, x2 + m2} + #) & /@ hex] }]];
PolyPlot[{Δ_, θ_}] := PolyPlot[θ]
```

```
☺ GraphicsRow[PolyPlot[θ[Knot[#]]] &
    /@ {"3_1", "K11n34", "K11n42"}]
```



K11n34 K11n42



So θ detects knot mutation and separates the Conway knot K11n34 from the Kinoshita-Terasaka knot K11n42!



Conway Kinoshita Terasaka

```
☺ GraphicsRow[PolyPlot[θ[TorusKnot @@ #]] &
    /@ {{13, 2}, {17, 3}, {13, 5}, {7, 6}},
   Spacings → 0]
```



The 48-crossing Gompf-Scharlemann-Thompson knot [GST] is significant because it may be a counterexample to the slice-ribbon conjecture:



Gompf Scharlemann Thompson



```
☺ AbsoluteTiming@
  PolyPlot[
```

$$\theta[\mathrm{EPD}[X_{14,1}, \bar{X}_{2,29}, X_{3,40}, X_{43,4}, \bar{X}_{26,5}, X_{6,95}, X_{96,7}, X_{13,8}, \bar{X}_{9,28}, X_{10,41}, X_{42,11}, \bar{X}_{27,12}, X_{30,15}, \bar{X}_{16,61}, \bar{X}_{17,72}, \bar{X}_{18,83}, X_{19,34}, \bar{X}_{89,20}, \bar{X}_{21,92}, \bar{X}_{79,22}, \bar{X}_{68,23}, \bar{X}_{57,24}, \bar{X}_{25,56}, X_{62,31}, X_{73,32}, X_{84,33}, \bar{X}_{50,35}, X_{36,81}, X_{37,70}, X_{38,59}, \bar{X}_{39,54}, X_{44,55}, X_{58,45}, X_{69,46}, X_{80,47}, X_{48,91}, X_{90,49}, X_{51,82}, X_{52,71}, X_{53,60}, \bar{X}_{63,74}, \bar{X}_{64,85}, \bar{X}_{76,65}, \bar{X}_{87,66}, \bar{X}_{67,94}, \bar{X}_{75,86}, \bar{X}_{88,77}, \bar{X}_{78,93}]]]$$

{39.0193,  }

The torus knot $T_{22/7}$:



Last, a random 250 crossing knot (knot from N. Dunfield; more at ωεβ/DK):



Prior Art. θ is probably equal to the "2-loop polynomial" studied by Ohtsuki at [Oh2] (at much greater difficulty, and with harder computations). θ is related, but probably not equivalent, to the invariant studied by Garoufalidis and Kashaev at [GK].



Ohtsuki Garoufalidis Kashaev

θ Sees Topology! Indeed, for a knot K, half the T_1 degree (say) of $\theta(K)$ bounds the genus of K from below, and this bound is sometimes better (and sometimes worse) than the bound coming from Δ. It is fair to hope that "anything Δ can do θ can do too" (see [BN2]), and in particular, that θ may say something about ribbon and/or slice properties.

The Rolfsen Table of Knots.



Where is it coming from? The most honest answer is "we don't know" (and that's good!). The second most, "undetermined coefficients for an ansatz that made sense". The ansatz comes from the following principles / earlier work:

Morphisms have generating functions. Indeed, there is an isomorphism

$$\mathcal{G}\colon\ \mathrm{Hom}(\mathbb{Q}[x_i],\mathbb{Q}[y_j]) \to \mathbb{Q}[y_j][\![\xi_i]\!],$$

and by PBW, many relevant spaces are polynomial rings, though only as vector spaces.

Composition is integration. Indeed, if $f\in\mathrm{Hom}(\mathbb{Q}[x_i],\mathbb{Q}[y_j])$ and $g\in\mathrm{Hom}(\mathbb{Q}[y_j],\mathbb{Q}[z_k])$, then

$$\mathcal{G}(g\circ f)=\int e^{-y\cdot\eta}fg\,dy\,d\eta$$

Use universal invariants. These take values in a universal enveloping algebra (perhaps quantized), and thus they are expressible as long compositions of generating functions. See [La, Oh1].

"Solvable approximation" $\rightsquigarrow$ perturbed Gaussians. Let $\mathfrak{g}$ be a semisimple Lie algebra, let $\mathfrak{h}$ be its Cartan subalgebra, and let $\mathfrak{b}^u$ and $\mathfrak{b}^l$ be its upper and lower Borel subalgebras. Then $\mathfrak{b}^u$ has a bracket β, and as the dual of $\mathfrak{b}^l$ it also has a cobracket δ, and in fact, $\mathfrak{g}\oplus\mathfrak{h}\equiv\mathrm{Double}(\mathfrak{b}^u,\beta,\delta)$. Let $\mathfrak{g}^+_\epsilon\coloneqq\mathrm{Double}(\mathfrak{b}^u,\beta,\epsilon\delta)$ (mod ϵ^{d+1} it is solvable for any d). Then by [BV3, BN1] (in the case of $\mathfrak{g}=sl_2$) all the interesting tensors of $\mathcal{U}(\mathfrak{g}^+_\epsilon)$ (quantized or not) are perturbed Gaussian with perturbation parameter ϵ with with understood bounds on the degrees of the perturbations.

The Philosophy Corner. "Universal invariants", valued in universal enveloping algebra (possibly quantized) rather than in representations thereof, are a priori better than the representation theoretic ones. They are compatible with strand doubling (the Hopf coproduct), and as the knot genus and the ribbon property for knots are expressible in terms of strand doubling, universal invariants stand a chance to say something about these properties. Indeed, they sometimes do! See e.g. [BN2, Oh2, GK, LV, BG]. Representation theoretic invariants don't do that!

References.

[BN1] D. Bar-Natan, *Everything around sl^ϵ_{2+} is DoPeGDO. So what?,* talk given in "Quantum Topology and Hyperbolic Geometry Conference", Da Nang, Vietnam, May 2019. Handout and video at ωεβ/DPG.

[BN2] D. Bar-Natan, *Algebraic Knot Theory,* talk given in Sydney, September 2019. Handout and video at ωεβ/AKT.

[BN3] D. Bar-Natan, *Cars, Interchanges, Traffic Counters, and some Pretty Darned Good Knot Invariants,* talk given in "Using Quantum Invariants to do Interesting Topology", Oaxaca, Mexico, October 2022. Handout and video at ωεβ/Cars.

[BV1] D. Bar-Natan and R. van der Veen, *A Polynomial Time Knot Polynomial,* Proc. Amer. Math. Soc. **147** (2019) 377–397, arXiv:1708.04853.

[BV2] D. Bar-Natan and R. van der Veen, *A Perturbed-Alexander Invariant,* to appear in Quantum Topology, ωεβ/APAI.

[BV3] D. Bar-Natan and R. van der Veen, *Perturbed Gaussian Generating Functions for Universal Knot Invariants,* arXiv:2109.02057.

[BG] J. Becerra Garrido, *Universal Quantum Knot Invariants,* Ph.D. thesis, University of Groningen, ωεβ/BG.

[GK] S. Garoufalidis and R. Kashaev, *Multivariable Knot Polynomials from Braided Hopf Algebras with Automorphisms,* arXiv:2311.11528.

[GST] R. E. Gompf, M. Scharlemann, and A. Thompson, *Fibered Knots and Potential Counterexamples to the Property 2R and Slice-Ribbon Conjectures,* Geom. and Top. **14** (2010) 2305–2347, arXiv:1103.1601.

[GPV] M. Goussarov, M. Polyak, and O. Viro, *Finite type invariants of classical and virtual knots,* Topology **39** (2000) 1045–1068, arXiv:math.GT/9810073.

[La] R. J. Lawrence, *Universal Link Invariants using Quantum Groups,* Proc. XVII Int. Conf. on Diff. Geom. Methods in Theor. Phys., Chester, England, August 1988. World Scientific (1989) 55–63.

[LV] D. López Neumann and R. van der Veen, *Genus Bounds from Unrolled Quantum Groups at Roots of Unity,* arXiv:2312.02070.

[Oh1] T. Ohtsuki, *Quantum Invariants,* Series on Knots and Everything **29**, World Scientific 2002.

[Oh2] T. Ohtsuki, *On the 2–Loop Polynomial of Knots,* Geom. Topol. **11-3** (2007) 1357–1475.

[Ov] A. Overbay, *Perturbative Expansion of the Colored Jones Polynomial,* Ph.D. thesis, University of North Carolina, August 2013, ωεβ/Ov.

[R1] L. Rozansky, *A Contribution of the Trivial Flat Connection to the Jones Polynomial and Witten's Invariant of 3D Manifolds, I,* Comm. Math. Phys. **175-2** (1996) 275–296, arXiv:hep-th/9401061.

[R2] L. Rozansky, *The Universal R-Matrix, Burau Representation and the Melvin-Morton Expansion of the Colored Jones Polynomial,* Adv. Math. **134-1** (1998) 1–31, arXiv:q-alg/9604005.

[R3] L. Rozansky, *A Universal $U(1)$-RCC Invariant of Links and Rationality Conjecture,* arXiv:math/0201139.

[Sch] S. Schaveling, *Expansions of Quantum Group Invariants,* Ph.D. thesis, Universiteit Leiden, September 2020, ωεβ/Scha.